

Using formulas to find Calc data

Andrew Douglas Pitonyak

Calc offers numerous methods to find data in a sheet. For example, *Edit -> Find & Replace* moves the display cursor based on simple and advanced searching. Use *Data -> Filter* to limit what is displayed rather than simply moving the cursor. Calc also offers lookup functions used in formulas. For example, a formula to lookup a student's grade based on their test score.

Search a block of data using VLOOKUP

Use VLOOKUP to search the first column (columns are vertical) of a block of data and return the value from a column in the same row. For example, search the first column for the name "Fred" and then return value in the cell two columns to the right. VLOOKUP supports two forms:

```
VLOOKUP(search_value; search_range;
return_column_index)
VLOOKUP(search_value; search_range;
return_column_index; sort_order)
```

The first argument, `search_value`, identifies the value to find. The search value can be text, a number, or a regular expression. For example, "Fred" searches for the text Fred, 4 searches for the number 4, and "F.*" is the regular expression for find something that starts with the letter F.

The second argument, `search_range`, identifies the cells to search; only the first column is searched. For example, B3:G10 searches the same sheet containing the VLOOKUP formula and Sheet2.B3:G10 searches the range B3:G10 on the sheet named Sheet2.

The `return_column_index` identifies the column to return; a value of 1 returns the first column in the range. The statement `=VLOOKUP("Bob"; A1:G9; 1)` finds the first row in A1:G9 containing the text "Bob", and returns the value in the first column. The first column is the searched column, so the

text Bob is returned. If the column index is 2, then the value in the cell to the right of Bob is returned; column B.

The final column, `sort_order`, is optional. The default value for `sort_order` is 1, which specifies that the first column is sorted in ascending order; a value of 0 specifies that the data is not sorted. A non-sorted list is searched by sequentially checking every cell in the first column for an exact match. If an exact match is not found, the text "#N/A" is returned.

A more efficient search routine is used if the data is sorted in ascending order. If one exact match exists, the returned value is the same as for a non-sorted list; but it is faster. If a match does not exist, the largest value in the column that is less than or equal to the search value is returned. For example, Searching for 7 in (3, 5, 10) returns 5 because 7 is between 5 and 10. Searching for 27 returns 10, and searching for 2 returns "#N/A" because there is no match and no value less than 2.

Use VLOOKUP when:

- The Data is arranged in rows and you want to return data from the same row. For example, student names with test and quiz scores to the right of the student's name.
- Searching the first column of a range of data.

Search a block of data using HLOOKUP

Use HLOOKUP to search the first row (rows are horizontal) of a block of data and return the value from a row in the same column. HLOOKUP supports the same form and arguments as VLOOKUP:

```
HLOOKUP(search_value; search_range;
return_row_index)
HLOOKUP(search_value; search_range;
return_row_index; sort_order)
```



Use HLOOKUP when:

- The Data is arranged in columns and you want to return data from the same column. For example, student names with test and quiz scores underneath the student's name.
- Searching the first row of a range of data.

Search a row or column using LOOKUP

LOOKUP is similar to HLOOKUP and VLOOKUP. The search range for the LOOKUP function is a single sorted row or column. LOOKUP has two forms:

```
LOOKUP(search_value; search_range)
```

```
LOOKUP(search_value; search_range; return_range)
```

The search value is the same as HLOOKUP and VLOOKUP. The search range, however, must be a single row or a single column; for example, A7:A12 (values in column A) or C5:Q5 (values in row 5). If the return_range is omitted, the matched value is returned. Functionally, using LOOKUP without a return range is the same as using HLOOKUP or VLOOKUP with a column index of 1.

The return range must be a single row or column containing the same number of elements as the search range. If the search value is found in the fourth cell in the search range, then the value in the fourth cell in the return range is returned. The return range can have a different orientation than the search range. In other words, the search range can be a row and the return range may be a column.

Use LOOKUP when:

- The search data is sorted in ascending order.
- The search data is not stored in the same row, column, or orientation as the return data.

Use MATCH to find the index of a value in a range

- Use MATCH to search a single row or column and return the position that matches the search value. Use MATCH to find the index of a value in a range. The supported forms for MATCH are as follows:

```
=MATCH(search_value; search_range)
```

```
=MATCH(search_value; search_range; search_type)
```

- The search value and search range are the same as for LOOKUP. The final argument, search type, controls how the search is performed. A search type of 1, sorted in ascending order, is the default. A search type of -1 indicates that the list is sorted in descending order. A search type of 0 indicates that the list is not sorted. Regular expressions can only be used on an unsorted list.

Use MATCH when:

- You need an index into the range rather than the value.
- The search data is in descending order and the data is large enough that the data must be searched assuming that it is sorted; because it is faster to sort a sorted list.

Examples

Consider the data in Table 1. Each student's information is stored in a single row. Write a formula to return the average grade for Fred. The problem can be restated as Search

Table 1. Simple grading spreadsheet

	A	B	C	D	E	F	G
1	Name	Test 1	Test 2	Quiz 1	Quiz 2	Average	Grade
2	Andy	95	93	93	92	93,25	
3	Betty	87	92	65	73	79,25	
4	Bob	95	93	93	92	93,25	
5	Brandy	45	65	92	85	71,75	
6	Frank	95	93	85	92	91,25	
7	Fred	87	92	65	73	79,25	
8	Ilsab	70	85	97	79	82,75	
9	James	45	65	97	85	73	
10	Lisa	100	97	100	93	97,5	
11	Michelle	100	97	100	65	90,5	
12	Ravi	87	92	86	93	89,5	
13	Sal	45	65	100	92	75,5	
14	Ted	100	97	100	85	95,5	
15	Tom	70	85	93	65	78,25	
16	Whil	70	85	93	97	86,25	

column A in the range A1:G16 for Fred and return the value in column F (column F is the sixth column). The obvious solution is `=VLOOKUP("Fred"; A2:G16; 6)`. Equally obvious, is `=LOOKUP("Fred"; A2:A16; F2:F16)`.

It is common for the first row in a range to contains column headers. All of the search functions check the first row to see if there is a match and then ignore it if it does not contain a match; in case the first row is a header.

What if the column heading “Average” is known, but not the column containing the average? Find the column containing “Average” rather than hard coding the value 6. A slight modification using MATCH to find the column yields `=VLOOKUP("Fred"; A2:G16; MATCH("Average"; A1:G1; 0))`; notice that the heading is not sorted. As an exercise, use HLOOKUP to find “Average” and then MATCH to find the row containing Fred.

As a final example, write a formula to assign grades based on a student’s average score. Assume that a score less than 51 is an F, less than 61 is an E, less than 71 is a D, less than 81 is a C, less than 91 is a B, and 91 to 100 is an A. Assume that the values in Table 2 are in Sheet2.

The formula `=VLOOKUP(83; $Sheet2.$A$2:$B$7; 2)` is an obvious solution. Dollar signs are used so that the formula can be copied and pasted to a different location and it will still reference the same values in Table 2.

Table 2. Associate scores to a grade

	A	B
1	Score	Grade
2	0	F
3	51	E
4	61	D
5	71	C
6	81	B
7	91	A




Figure 1. List Box Wizard – Data dialog

Advanced example using controls

We use a program that reads an input file of comma separated data. Most people edit the data in a text editor, but it is easy to make errors. I wrote macros to read and write the data from a Calc document, which is configured in an easy to use and understandable format.

Learn to use a list box

One field requires the values YES or NO. Use a List box to set the value YES or NO. We use a list box configured as a drop down to limit the entered values. We use a list box configured as a drop down rather than a drop down, because a drop down control allows the user to enter a value in the text edit box that is not in the list – and that is not acceptable.

Use *View -> Toolbars* to verify that there is a check next to the Form Controls toolbar. Click the design mode on/off icon to turn on the form design mode. With design mode on, you can add and modify controls. With design mode off, you can use the controls added to the sheet. The list box icon changes from disabled to enabled. Click on the list box icon and the cursor changes to a cross shape. Click in the Calc sheet and drag a rectangle that defines the size of the list box over cell D16 on Sheet1; the List Box Wizard – Data dialog is displayed (see Figure 1).

Click *Cancel* to close the dialog and configure the control manually. The cursor is still a cross indicating




Figure 2. Properties dialog for a list box control

that another list box can be added. Click the select icon in the *Form Controls* toolbar to change the cursor back to normal. Right click on the control and choose *Control* to open the Properties dialog (see Figure 2).

- Set the control name to FlipListBox.
- Place the cursor in the List entries box. Type the word NO, press [Shift+Enter], type the word YES, and press [Enter]. [Shift+Enter] separates text entries and [Enter] finishes entering a value. The list box now contains the values YES and NO.
- Use the Default selection to set the default value. Click the icon to the right of the property to open the Default selection dialog (see Figure 3). Select NO and click OK.



- Set the Help text property to “A value of YES causes the Z-axis to be flipped during the calculation.” The Help text property sets the help text displayed in user mode when the mouse is moved over the control.
- Click the *Data* tab and set the Linked cell to D14 (see Figure 4).

Click the design mode on/off icon to turn off the form design mode. Test the list box. When the value changes in the list box, the value changes in cell D14. Also, enter the text YES or NO in cell D14, and the value in the list box also changes. Hover the cursor over the list box and the help text is displayed.

Use a list box and a search

Another field indicated how an integral is calculated (something from numerical methods and Calculus). Three numerical values 0, 1, or 2, correspond to the Monte Carlo method, Simpson’s rule, or Trapezoidal rule. The working document contains lookup tables in Sheet2 and displays the entry from in Sheet1. To more easily see the results, enter values in Sheet1.E20:F23 (see Table 3).

Add and configure a new list box over cell E15.

- 1 Turn design mode on.
- 2 Click the list box icon.
- 3 Draw a list box over cell E15.
- 4 The List Box Wizard – Data dialog is displayed (see Figure 1); click *Cancel*.
- 5 Right click on the list box and select *Control*.
- 6 Click the *Data* tab.
- 7 Set the linked cell to F20.
- 8 Set the Source cell range to E21:E23, which corresponds to the values in Table 3.
- 9 Select the *General* tab.

	E	F
20	Integration	
21	Monte Carlo	0
22	Simpson	1
23	Trapezoidal	2

Table 3. Supported integration methods



Figure 3. Select the list box default value




Figure 4. Set the Linked cell for the list box

- 10 Set the control name to IntegrationListBox.
- 11 The List entries property is read-only and is set to Monte Carlo, Simpson, and Trapezoidal.
- 12 Set the Default selection to Simpson, which displays a “1” for the property.
- 13 Set the Help text to “Choose an integration method.”
- 14 Turn off the design mode.

Enter the formula =VLOOKUP(F20; E21:F23; 2; 0) in cell E14. The list box is linked to cell F20, which

means that it contains the desired text. The formula in cell E14 uses the text in cell F20 to lookup the correct numerical value in Table 3. To make the sheet more user friendly, right click on cell E14 and select Insert Note. Enter the text “0=Monte Carlo, 1=Simpson, and 2=Trapezoidal”. A red dot is displayed in the upper right corner of cell E14 to indicate that there is an attached note. Hover the cursor over cell E14 to display the note (see Figure 6).

Figure 5. Set the Linked cell for the list box





Figure 6. A red dot indicates an attached note

Conclusion

Searching and filters are primarily used to find and view a small subset of data from a larger set – find all students with an A. Formulas, on the other hand, are used to lookup data used in calculations and then set a value in a cell – assign a grade to a student based their score. Hopefully these examples will help you choose the appropriate formula for your Calc documents. ■